query: SELECT ?item ?itemLabel 
    ?ct (ROUND( 100 * ?ct / ?all ) as ?pc_all)  ?all 
    ?ct1 (ROUND( 100 * ?ct1 / ?allunique ) as ?pc_allunique) ?allunique
    ?currentyrct
    ?lastyrct
    ?sample
  WITH
  {  SELECT (COUNT( DISTINCT ?a ) as ?all) { ?a wdt:P31 wd:Q47150325 ; wdt:P585 []  } } as %inc
  WITH
  {  SELECT (COUNT( DISTINCT ?a1 ) as ?allunique) { ?a1 wdt:P31 wd:Q47150325 ; wdt:P585 [] . { ?a1 wikibase:sitelinks 1 } UNION { ?a1 wikibase:sitelinks 0 }  } } as %inc2
  WHERE
  {
      INCLUDE %inc 
      INCLUDE %inc2            
      {
      {
          SELECT ?wiki (COUNT(DISTINCT ?lh) as ?ct) (SUM(?currentyr) as ?currentyrct) (SUM(?lastyr) as ?lastyrct) (SAMPLE(?lh) as ?sample)
          WHERE
          {
      	  ?lh wdt:P31 wd:Q47150325 ; wdt:P585 ?d .
      	  [] schema:about ?lh ; schema:isPartOf ?wiki
            BIND(IF(YEAR(?d)=YEAR(Now()),1,0) as ?currentyr )
            BIND(IF(YEAR(?d)=YEAR(Now())-1,1,0) as ?lastyr )                          
          }
  	    GROUP BY ?wiki
          HAVING(?ct > 14 )
  	}    
  	OPTIONAL {
          SELECT ?wiki (COUNT(DISTINCT ?lh) as ?ct1)
          WHERE
          {
      	  ?lh wdt:P31 wd:Q47150325 ; wdt:P585 [] .
            ?lh wikibase:sitelinks 1 .
      	  [] schema:about ?lh ; schema:isPartOf ?wiki
          }
  	    GROUP BY ?wiki
          HAVING(?ct1 > 0 )
  	}
  	?item wdt:P856 ?wiki .
      } UNION { SELECT (wd:Q31747418 as ?item) (COUNT( DISTINCT ?b ) as ?ct) (?ct as ?ct1) { ?b wdt:P31 wd:Q47150325 ; wikibase:sitelinks 0 ; wdt:P585 [] } } 
      SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
  }
  ORDER BY DESC (?currentyrct)
  LIMIT 30